

Empirical algorithms to restore a complete set of inherent optical properties of seawater

Vladimir I. Haltrin

Naval Research Laboratory, Ocean Sciences Branch, Code 7331
Stennis Space Center, MS 39529-5004, USA, e-mail: <haltrin@nrlssc.navy.mil>

ABSTRACT

This work analyzes available experimental and in situ information obtained by different investigators and presents it in the form of simple usable algorithms and codes. The following independent algorithms are presented: 1) the retrieval of backscattering coefficient through the measurements of beam attenuation and beam scattering coefficients, 2) the retrieval of angular scattering coefficient through the measurements of beam attenuation and absorption coefficients, 3) the retrieval of downward and upward mean cosines in the depth regime through the Gordon's parameter that depends on backscattering and absorption coefficient, 4) the retrieval of diffuse attenuation and diffuse reflection coefficients through the Gordon's parameter. The applicability of these algorithms is discussed and examples of their applications are presented.

Keywords: Ocean optics, inherent optical properties, scattering, absorption, seawater.

1.0 INTRODUCTION

A complete set of inherent optical properties of the seawater includes a beam attenuation and an angular scattering coefficients. Due to financial and/or technological limitations majority of *in situ* optical experiments does not include a complete set of hydrooptical measurements. To use these experimental data to enhance sea water optical models and calibrate atmospheric correction algorithms it is necessary to have some means to estimate the absent inherent optical properties through the measured ones.

The algorithms presented here allow us to predict all inherent optical properties, including angular scattering coefficient $\beta(\theta)$ (here θ is a scattering angle), from the values of any two pair of the following optical properties: scattering coefficient b , absorption coefficient a , beam attenuation coefficient $c = a + b$, and single scattering albedo $\omega_0 = b / c$. These algorithms are based on an experimental data and does not include spectral dependence on wavelength. The algorithm to restore spectral inherent optical properties may be found in another paper published in these proceedings (Haltrin, 1999b).

2.0 ALGORITHM TO RETRIVE DOWNWARD AND UPWARD MEAN COSINES

Theoretically average cosine over radiance distribution in the depth of scattering media, and downward and upward cosines over the same distribution, depend on two optical properties: single-scattering albedo $\omega_0 = b/c$ and phase function of scattering $p(\cos\theta)$ (see Zege, Ivanov and Katsev, 1991; Haltrin, 1997a and 1997b, Haltrin, Kattawar and Weidemann, 1997; Haltrin 1998). But under natural maritime conditions some kind of biological equilibrium is usually established. This equilibrium explains significant correlation between inherent optical properties (Timofeyeva, 1971a and 1971b; Petzold, 1972; Efimenko and Pelevin, 1975; Haltrin, 1985; Aas, Højerslev, and Lundgren, 1997; and Haltrin, 1998a). Based on analysis of experimental data Haltrin (1997b) proposed the following regression between average cosine $\bar{\mu}$ and single scattering albedo:

$$\bar{\mu} = \sqrt{1-\omega_0} \sum_{n=0}^6 c_n (1-\omega_0)^{\frac{n}{2}} \equiv \sqrt{1-\frac{b}{c}} \sum_{n=0}^6 c_n \left(1-\frac{b}{c}\right)^{\frac{n}{2}}. \quad (1)$$

Connections between downward and upward average cosines (μ_d and μ_u) and average cosine $\bar{\mu}$ were proposed earlier in the paper by Haltrin and Weidemann (1996). These empirical relationships represent the experimental data collected by Timofeyeva (1971-1979) and other investigators, and they have the following analytic form:

$$\mu_d = \frac{1-\bar{\mu}(1-\bar{\mu})^2 \sum_{n=0}^3 d_n \bar{\mu}^{2n}}{2-\bar{\mu}}, \quad (2)$$

$$\mu_u = \frac{1-\bar{\mu}(1-\bar{\mu})^2 \exp\left(\sum_{n=0}^4 u_n \bar{\mu}^{2n}\right)}{2-\bar{\mu}}. \quad (3)$$

The coefficients c_n , d_n , and u_n in Eqs. (1)-(3) are given in Table 1.

Table 1. Coefficients to Eqs. (1)-(3).

n	c_n	d_n	u_n
0	2.6178398	0.0326	-0.0131
1	-4.6024180	0.1661	8.4423
2	9.0040600	0.7785	-15.6605
3	-14.59994	0.0228	21.8820
4	14.83909		-11.2257
5	-8.117954		
6	1.8593222		

Equations (1)-(3) express all three average cosines through the value of a single scattering albedo. These relationships within 10% of accuracy coincide with the independently developed one-parameter model of seawater optical properties (see Haltrin, 1998b and 1999b).

3.0 ALGORITHM TO RETRIVE BACKSCATTERING COEFFICIENT

Remote sensing reflectance and other optical properties of the seawater may be obtained from two basic optical properties: absorption coefficient a and backscattering coefficient b_B (Haltrin, 1984; Gordon, Brown and Jackobs, 1975; Gordon, 1993).

The relationship between backscattering coefficient and absorption and beam attenuation may be obtained from the following chain of equations: 1) reversal of the Gordon's parameter definition, $g = b_B / (a + b_B)$:

$$b_B = \frac{g a}{1 - g} \equiv \frac{g(c - b)}{1 - g}, \quad (4)$$

2) dependence of Gordon's parameter g on average cosine $\bar{\mu}$ (Haltrin and Kattawar, 1993; Haltrin, 1998a):

$$g = \frac{(1 - \bar{\mu}^2)^2}{1 + \bar{\mu}^2 (4 - \bar{\mu}^2)}, \quad (5)$$

3) Equation (1) that connects average cosine with absorption and attenuation coefficients.

4.0 ALGORITHM TO RETRIVE DIFFUSE REFLECTION AND DIFFUSE ATTENUATION COEFFICIENTS

The diffuse reflection R coefficient is determined through the well-known relationship:

$$R = \frac{1 - \bar{\mu} / \bar{\mu}_d}{1 + \bar{\mu} / \bar{\mu}_u} \equiv \frac{\bar{\mu}_u}{\bar{\mu}_d} \cdot \frac{\bar{\mu}_d - \bar{\mu}}{\bar{\mu}_u + \bar{\mu}}, \quad (6)$$

which connects R with average cosines $\bar{\mu}$, $\bar{\mu}_d$ and $\bar{\mu}_u$ (Eqs. (1)-(3)). Equation (6) may be easily derived by integrating a radiative transfer equation (Haltrin, 1984; Zege, Ivanov and Katsev, 1991).

The asymptotic diffuse attenuation coefficient k_∞ is determined through the Gershun's equation, $k_\infty = a / \bar{\mu}$, or, after a substitution of Eq. (1):

$$k_\infty = \sqrt{c^2 - bc} / \sum_{n=0}^6 c_n [1 - (b/c)]^n. \quad (7)$$

The downward and upward depth dependent diffuse attenuation coefficients under arbitrary surface illumination may be determined by Eqs. (42) and (43) in Haltrin (1998b).

5.0 ALGORITHM TO RETRIVE SEAWATER LIGHT SCATTERING PHASE FUNCTION

Empirical representation of angular scattering coefficient $\beta(\theta)$ and of the scattering phase function $p(\theta)$ based on experimental data by Petzold (1972) is derived in Haltrin, (1997b):

$$\beta(\theta) = \exp \left[q \sum_{n=0}^5 k_n \theta^{\frac{n}{2}} \right], \quad (8)$$

$$p(\theta) = \frac{4\pi}{b} \exp \left[q \sum_{n=0}^5 k_n \theta^{\frac{n}{2}} \right], \quad (9)$$

$$\frac{1}{2} \int_0^\pi p(\theta) \sin \theta d\theta = 1, \quad (10)$$

here θ is the scattering angle in degrees and coefficients q and k_n ($n = 0, 5$) are represented by the following equations:

$$q = 2.598 + 17.748\sqrt{b} - 16.722b + 5.932b\sqrt{b}, \quad (11)$$

$$\left. \begin{aligned} k_0 &= 1, & k_2 &= 0.307 - 0.19\omega_0, & k_4 &= 10^{-3}(3.24 - 2.25\omega_0), \\ k_1 &= 0.688\omega_0 - 1.188, & k_3 &= 0.0302\omega_0 - 0.0458, & k_5 &= 10^{-4}(0.61\omega_0 - 0.84). \end{aligned} \right\} \quad (12)$$

The regressions given by Eqns. (8)-(12) can be used as a basis for an empirical model of the seawater phase function with the coefficients dependent on the absorption and scattering coefficients. The single-scattering albedo used here varies from 0.09 to 0.96.

6.0 CONCLUSION

The set of equations (1)-(12) presented here allow us to calculate the whole set of inherent optical properties from two input parameters: absorption coefficient b and beam scattering coefficient c . This model is compared with independently derived one-parameter model of sea optical properties (Haltrin, 1998c; Haltrin, 1999). The predictions of this model coincide with the predictions of the one-parameter model with the error about 10%. The full working FORTRAN code of this model is given in the APPENDIX of this paper. The output of this code includes backscattering coefficient b_B , probability of backscattering B , all average cosines, diffuse and attenuation coefficients, and a phase function of scattering. The results of a sample output are shown in Fig.1.

7.0 ACKNOWLEDGMENT

The author thanks continuing support at the Naval Research Laboratory through the programs SS 5939-A9 and LOE 6640-09. This article represents NRL contribution PP/7331-98-0037.




Fig. 1. Examples of output Phase Functions

Table 2. Results of three sample runs of the program fiopeexp.

Input:			
b	0.30	0.5	1.0
c	0.45	1.2	3.0
Output:			
a	0.15	0.70	2.0
b_B	0.007990	0.009431	0.016091
B	0.026633	0.018862	0.016091
$\bar{\mu}$	0.778909	0.885248	0.910921
$\bar{\mu}_d$	0.805688	0.890244	0.913432
$\bar{\mu}_u$	0.350256	0.618432	0.721439
g	0.050572	0.013294	0.007981
R	0.010310	0.002308	0.001215
k_∞	0.192577	0.790739	2.195581

APPENDIX: A PROGRAM TO CALCULATE ALL INHERENT OPTICAL PROPERTIES THROUGH A SCATTERING AND A BEAM ATTENUATION COEFFICIENT

```
! program fiopeexp; written by Vladimir I. Haltrin, <haltrin@nrlssc.navy.mil>
! this program is free for any non-commercial use
! a reference to this article is required
implicit none
integer i,Nang,Nan
parameter (Nang = 361)
real*8 b,c,a,g,bB,Bbk,R,mu,mud,muu,kdf
real*8 ang(Nang),phf(Nang)
character tb
logical badinput

open(11, file='fiopeexp.in', status='old')
read(11,*) b
read(11,*) c
read(11,*) Nan ! Nan < Nang
read(11,*) (ang(i), i=1,Nan)
close(11)

if ((c.le.0.).or.(b.lt.0.).or.(c.lt.b)) then
badinput = .true.
else
badinput = .false.
call siopexp(b,c, a,bB,Bbk,mu,mud,muu,g,R,kdf)
call sphf(b,c,Nan,ang, phf)
end if
tb = CHAR(9)
```

```
open(21, file='fiopep.out', status='new')
if (badinput) then
  write(21,'(a)') 'bad input in "fiopep.in"'
else
  write(21,30) 'Input:          Scattering coefficient, b = ',b
&
  write(21,30) 'Input:          Attenuation coefficient, c = ',c
&
  write(21,*)
  write(21,'(a7)') 'Output:'
  write(21,30) 'Computed absorption coefficient, a = ',a,' 1/m'
  write(21,30) 'Computed backscattering coefficient, bB = ',bB
&
  write(21,30) 'Computed backscattering probability, B = ',bBk
  write(21,30) 'Computed average cosine <mu> = ',mu
  write(21,30) 'Computed downward average cosine <mud> = ',mud
  write(21,30) 'Computed upward average cosine <muu> = ',muu
  write(21,30) "Computed Gordon's parameter g = ",g
  write(21,30) 'Computed diffuse reflectance coefficient, R = ',R
  write(21,30) 'Computed diffuse attenuation coefficient, k = '
&
  open(22,file='phf.out', status='new')
  write(22,40) 'ang, °',tb,'phfunc'
  do i=1,Nan
    write(22,50) ang(i),tb,phf(i)
  end do
  close(22)
end if
close(21)

30 format(a42, f10.6,a7)
40 format(a6,a1,a6)
50 format(f6.2,a1,g12.5)

end

subroutine sphf(b,c,Nan,ang, phf)
implicit none
integer i,Nan
real*8 b,c,ang(Nan),phf(Nan)
real*8 q,omg,p,sqa,sqb,k1,k2,k3,k4,k5
sqb = SQRT(b)
omg = b/c
q = 2.598+sqb*(17.748+sqb*(5.932*sqb-16.722))
do i=1,Nan
  sqa = SQRT(ang(i))
  k1 = 0.688*omg - 1.188
  k2 = 0.307 - 0.19*omg
  k3 = 0.0302*omg - 0.0458
  k4 = (3.24 - 2.25*omg)*1.e-3
  k5 = (0.61*omg - 0.84)*1.e-4
  p = 1.+sqa*(k1+sqa*(k2+sqa*(k3+sqa*(k4+k5*sqa))))
  phf(i) = (12.56637062/b)*EXP(q*p)
end do
return
end
```

```

subroutine siopexp(b,c, a,bB,Bbk,mu,mud,muu,g,r,kdf)
implicit none
real*8 b,c, a,bB,Bbk,mu,mud,muu,g,r,kdf
real*8 fmuav,fgord
a = c-b
mu = fmuav(b,c)
g = fgord(mu)
bB = g*a/(1.-g)
Bbk = bB/b
call fsememp(mu, mud,muu,r)
kdf = a/mu
return
end

real*8 function fmuav(b,c)
implicit none
real*8 b,c,y
y = SQRT(1.-b/c)
fmuav = y*(2.6178398+y*(-4.6024180+y*(9.0040600+
& y*(-14.59994+y*(14.83909+y*(-8.117954+1.8593222*y))))))
return
end

real*8 function fgord(mu)
implicit none
real*8 mu,m2,g
m2 = mu*mu
g = (1.-m2)
fgord = g*g/(1.+m2*(4.-m2))
return
end

subroutine fsememp(mu, mud,muu,R)
implicit none
real*8 mu,mud,muu,R,m2,d,z
m2 = mu*mu
d = 1.-mu-mu+m2
z = 1./(2.-mu)
mud = (1.-mu*d*(0.0326+m2*(0.1661+m2*(0.7785+0.0228*m2))))*z
muu = (1.-mu*d*exp(-.0131+m2*(8.4423+m2*(-15.6605
& + m2*(21.882-11.2257*m2)))))*z
R = (1.-mu/mud)/(1.+mu/muu)
return
end
    
```

Input file "fiopexp.in":

```

0.3      <-- scattering coefficient b in 1/m (0 < b <= c)
0.45     <-- beam attenuation coefficient c in 1/m
30      <-- Nan, a number of angles for phase function (in degrees); ang(Nan):
0.      0.1  0.3  0.5  1.   1.5  3.0  5.   10  15.
20.     25.  30.  35.  40.  45.  50.  60.  70.  80.
90.     100. 110. 120. 130. 140. 150. 160. 170. 180.
    
```

REFERENCES

- E. Aas, N. K. Højerslev, and B. Lundgren, *Spectral irradiance, radiance and polarization data from the Nordic cruise in the Mediterranean Sea during June-July 1971*, (Report No. 102, Institutt for Geofysikk, Universitet i Oslo, September 1997).
- I. D. Efimenko and V. N. Pelevin. "Angular distribution of solar radiation in the Indian Ocean," in *Geophysical and Optical Studies in the Indian Ocean* (in Russian), L. M. Brekhovskikh and K. S. Shifrin, eds. (Nauka, Moscow, 1975), pp. 124-132.
- H. R. Gordon, O. B. Brown, and M. M. Jacobs, "Computed relationships between the inherent and apparent optical properties of a flat homogeneous ocean," *Appl. Optics*, **14**, 417-427 (1975).
- H. R. Gordon, "Sensitivity of radiative transfer to small-angle scattering in the ocean: Quantitative assessment," *Appl. Optics*, **32**, 7205-7511 (1993).
- V. I. Haltrin (a.k.a. В. И. Халтурин), "Propagation of light in sea depth," Chapter 2 in: *Remote Sensing of the Sea and the Influence of the Atmosphere*, pp. 20-62 (in Russian), V. A. Urdenko and G. Zimmermann, eds., Moscow-Berlin-Sevastopol, Published by the German Democratic Republic Academy of Sciences Institute for Space Research, (1985).
- V. I. Haltrin and G. W. Kattawar "Self-consistent solutions to the equation of transfer with elastic and inelastic scattering in oceanic optics: I. Model," *Appl. Optics*, **32**, pp. 5356-5367, (1993).
- V. I. Haltrin, "Theoretical and empirical phase functions for Monte Carlo calculations of light scattering in seawater," in *Proceedings of the Fourth International Conference Remote Sensing for Marine and Coastal Environments*, **I**, pp. 509-518, Envir. Res. Inst. of Michigan, Ann Arbor, MI, USA, (1997a).
- V. I. Haltrin, "Retrieval of Remote Radiance Reflection Coefficients of Coastal Waters from the Inherent Optical Properties," in *Proceeding of the International Geoscience and Remote Sensing Symposium IGARSS'97*, Vol. I, pp. 595-597, Singapore (1997b).
- V. I. Haltrin, G. W. Kattawar, and A. D. Weidemann, "Modeling of elastic and inelastic scattering effects in oceanic optics," in *SPIE*, **2963**, S. G. Ackleson and R. Frouin, eds., *Ocean Optics XIII*, Bellingham, WA, pp. 597-602, (1997).
- V. I. Haltrin, "Self-consistent approach to the solution of the light transfer problem for irradiances in marine waters with arbitrary turbidity, depth and surface illumination" *Appl. Optics*, **37**, pp. 3773-3784 (1998a).
- V. I. Haltrin, "Apparent optical properties of the sea illuminated by Sun and sky: case of optically deep sea," *Appl. Optics*, **37**, 8336-8340, (1998b).
- V. I. Haltrin, "One-parameter model of seawater optical properties," in *Ocean Optics XIV CD-ROM*, Kailua-Kona, Hawaii, 10-13 November 1998, Published by Office of Naval Research, (1998c).
- V. I. Haltrin and A. D. Weidemann, "A Method and Algorithm of Computing Apparent Optical Properties of Coastal Sea Waters," in *Proceedings of International Geoscience and Remote Sensing Symposium: IGARSS'96*, **I**, pp. 305-309, Lincoln, Nebraska (1996).
- V. I. Haltrin, "Diffuse reflection coefficient of stratified sea," *Appl. Optics*, **38**, 932-936, (1999a).
- V. I. Haltrin, "An algorithm to restore spectral signatures of all inherent optical properties of seawater using a value of one property at one wavelength," in: *Proceedings of the Fourth International Airborne Remote Sensing Conference and Exhibition/21st Canadian Symposium on Remote Sensing*, pp. II-680-II-687, ISSN 1076-7924, Published by ERIM International, Inc., Ann Arbor, MI, USA (1999b).
- T. J. Petzold, *Volume Scattering Functions for Selected Ocean Waters*, SIO Ref. 72-78, Scripps Institute of Oceanography, Visibility Laboratory, San Diego, CA, (1972).
- V. A. Timofeyeva, "Optical characteristics of turbid media of the seawater type," *Izv. Atmos. Ocean Physics*, **7**, pp. 863-865, (1971a).
- V. A. Timofeyeva, "The diffuse reflection coefficient and its relation to the optical parameters of turbid media," *Izv. Atmos. Ocean Physics*, **7**, pp. 467-469, (1971b).
- V. A. Timofeyeva, "Determination of Light-Field parameters in the depth regime from irradiance measurements," *Izv. Atmos. Ocean Physics*, **15**, pp. 774-776, (1979).
- E. P. Zege, A. P. Ivanov and I. L. Katsev, *Image Transfer through a Scattering Media* (Springer Verlag, Berlin, 1991).